
Группа Пример к теме

4 Реверсивный переключатель для переключения
направления вращения асинхронного двигателя

Краткое описание

Данный пример программы служит для управления асинхронным двигателем с двума
направлениями вращения.

Двигатель может быть запущен на движение влево нажатием кнопки на входе E0.0 или вправо
нажатием кнопки на E0.1, при условии, что защитный выключатель на входе E0.3 и основной
выключатель на E0.2 не нажаты. Переключить направление вращение можно только после нажатия
основного выключателя и окончания времени ожидания 5 с, при этом может происходить
торможение и запуск двигателя. Если обе кнопки на включение нажаты одновременно, то двигатель
будет блокирован.

Схема включения

Асинхронн.
двигатель

Защитный
контактор

24 В

SIEMENS
CPU 212

SIMATIC
S7-200

RUN
STOP
SF

� � � � �¦ � � 	 �
� � �

Выкл ПР  ЛЕ  Ожидание

��� � � �¦ �� 	 �
�� �

XXX X

Напряжение на двигат.
L1    L2     L3

Контактор
Движение
вправо

Контактор
Движение
влево

24

Ле Пр Выкл.

  SIMATIC
S7-200 Примеры

Industrial automation
Elincom Group

European Union: www.elinco.eu
Russia: www.elinc.ru



S7-200 Примеры    Реверсивный переключатель (1)  Пример-Nо. 10

Copyright  1996 by SIEMENS Страница  1 / 6 S72_10r.doc
Выпуск: 10/96 Версия  3.1

Структура программы

Rechtslauf

AUS

Linkslauf

Taster E0.0 Taster E0.1

Aus-Schalter E0.2

Описание программы включая листинг

Сначала программа проверяет, должна ли быть включена блокировка установкой вспомогательного
меркера M2.0. Это как раз тот случай когда обе кнопки, на входах E0.0 и E0.1 нажаты одновременно
или если время ожидания еще не вышло. Только если обе кнопки снова находятся в отжатом
состоянии и кончилось время ожидания, т.е вспомогательный меркер M2.3 сброшен, будет сброшен
и блокирующий меркер M2.0.

Разрешение на вращение влево осуществляется установкой соответствующего бита деблокировки
M2.1, если ни защитный выключатель двигателя на входе E0.3 или кнопка ВЫКЛ на E0.2 (оба
нормально закрытые) не нажаты, ни бит состояния для вращения вправо M1.1 не установлен. Бит
состояния для вращения влево это M1.0. Деблокировка вращения вправо происходит аналогично.

Двигатель включается, если нажата одна кнопка включения и ни бит блокировки ни бит состояния
другого направления не установлены. Пуск двигателя происходит установкой соответствующего
выхода и относящегося к нему бита состояния, обеспечивающего самоподхват. На выходе A0.0
срабатывает контактор левого вращения, а на выходе A0.1 правого. Кроме того загорается
сигнальная лампа режима. Сигнальная лампа движения влево подключена на A0.4, движения
вправо на A0.3, а лампа состояния ВЫКЛ на A0.2.

Если двигатель выключается, то по положительному фронту ‘ED’ устанавливается вспомогательный
меркер M2.3 (процесс выключения). Если последний установлен, то запускается таймер ожидания
для повторного пуска со временем работы 5 с (500*10 мс). По завершении этого времени меркер
M2.3 процесса выключения сбрасывается. Во время работы таймера ожидания на выходе A0.5
мигает сигнальная лампа. В случае, если бит состояния не установлен, то на выходе A0.2 горит
сигнальная лампа состояния ‘ВЫКЛ’.

Размер программы составляет 61 слово.



S7-200 Примеры    Реверсивный переключатель (1)  Пример-Nо. 10

Copyright  1996 by SIEMENS Страница  1 / 6 S72_10r.doc
Выпуск: 10/96 Версия  3.1

KOP (S7-MicroDOS) AWL (TOOLITE2)

Основная программа

// TITEL = Контакторное переключение

// Блокировка

│ E0.1 E0.0 M2.0 K1
1 ├─┤ ├───┤ ├─┬─────────────────────────────────( S )

│ │
│ M2.3 │
├─┤ ├─────────┘

LD  E0.1 // Команда вправо
U  E0.0 // и команда влево
O  M2.3 // или время ожидания не кончилось
S  M2.0,1 // устанавливается блокировка

│ E0.0 E0.1 M2.3 M2.0 K1
2 ├─┤ / ├───┤ / ├───┤ / ├───────────────────────────( R )

│

LDN  E0.0 // Нет команды влево
UN  E0.1 // нет команды вправо
UN  M2.3 // время ожидания кончилось
R  M2.0, 1 // блокировка сбрасывается

// Деблокировка движения влево

│ E0.2 E0.3 M1.1 M2.1
3 ├─┤ ├───┤ ├───┤ / ├───────────────────────────( )

│

LD  E0.2 // Нет команды выключения
U  E0.3 // не сработал защитный выключатель
UN  M1.1 // бит состояния движения вправо не установлен
=  M2.1

// Деблокировка движения вправо

│ E0.2 E0.3 M1.0 M2.2
4 ├─┤ ├───┤ ├───┤ / ├───────────────────────────( )

│

LD  E0.2 // Нет команды выключения
U  E0.3 // не сработал защитный выключатель
UN  M1.0 // бит состояния движения влево не установлен
=  M2.2

// Движение влево

│ E0.0 M2.0 M2.1 M1.0
5 ├─┤ ├─┬─┤ / ├───┤ ├──────────────┬────────────( )

│ │ │
│ M1.0 │ │ A0.0
├─┤ ├─┘ ├────────────( )
│ │
│ │ A0.4
│ └────────────( )
│

LD  E0.0 // Команда влево
O  M1.0 // или состояние левого движения
UN  M2.0 // и нет блокировки
U  M2.1 // разрешение на движение влево
=  M1.0 // бит состояния движения влево
=  A0.0 // выход контактора
=  A0.4 // выход сигнальной лампы движения влево

// Движение вправо

│ E0.1 M2.0 M2.2 M1.1
6 ├─┤ ├─┬─┤ / ├───┤ ├──────────────┬────────────( )

│ │ │
│ M1.1 │ │ A0.1
├─┤ ├─┘ ├────────────( )
│ │
│ │ A0.3
│ └────────────( )
│

LD  E0.1 // Команда вправо
O  M1.1 // или состояние правого движения
UN  M2.0 // и нет блокировки
U  M2.2 // разрешение на движение вправо
=  M1.1 // бит состояния движения вправо
=  A0.1 // выход контактора
=  A0.3 // выход сигнальной лампы движения вправо

// Определение фронта процесса выключения

│ M1.0 M1.1 A0.2
7 ├─┤ / ├───┤ / ├───────────────────────────────────( )

│

LDN  M1.0 // бит состояния движения влево
UN  M1.1 // бит состояния движения вправо
=  A0.2 // выход сигнальной лампы состояния ВЫКЛ

│ A0.2 M2.3 K1
8 ├─┤ ├───┤ N ├───────────────────────────────────( S )

│

LD  A0.2 // При выключении
ED
S  M2.3,1 // Устанавливается вспомогательный меркер для 

         // процесса выключения

│ M2.3 MOV_W───┐
9 ├─┤ ├─┬────────────────────────────┬────────────┤EN │

│ │ │ │ │
│ │ │ K500┤IN OUT├VW20
│ │ │ │ │
│ │ │ └───────┘
│ │ │ TON───T33
│ │ └────────────┤IN │
│ │ │ │
│ │ VW20┤PT │
│ │ │ │
│ │ └───────┘
│ │ T33 M2.3 K1
│ └─┤ ├───────────────────────────────────( R )

LD  M2.3
MOVW  500,VW20 // Загрузка времени ожидания для перезапуска
TON  T33,VW20 // Пуск таймера ожидания для перезапуска
U  T33
R  M2.3, 1 // Сброс вспомогательного меркера после времени ожидания

// Индикация состояния ВЫКЛ, Время ожидания

│ M2.3 SM0.5 A0.5
10 ├─┤ ├───┤ ├───────────────────────────────────( )

│

LD  M2.3 // Вспомогательный меркер времени ожидания 
           // процесса выключения

U  SM0.5 // Мигание 1 с
=  A0.5 // Выход сигнальная лампа время ожидания

│
11 ├─────────────────────────────────────────────────( MEND )

│

MEND // Конец основной программы

Указания по преобразованию

Для того чтобы преобразовать TOOLITE2 AWL в S7-Micro/DOS AWL

            •   Установите ‘K’ перед каждым числом, не являющимся 16-ричной
      константой (напр. 4 → K4)

            •   Замените ‘16#’ на ‘KH’ для всех 16-ричных констант
      (напр. 16#FF → KHFF)

            •   Поставьте запятые для смены полей. Используйте клавиши
                 перемещения или клавишу TAB для перехода от поля к полю.
            •   Для преобразования программы S7-Micro/DOS AWL в KOP-форму нужно начинать каждый

сегмент словом ‘NETWORK’ и номером. Каждый сегмент в этом примере имеет свой
номер на диаграмме KOP. Используйте NWENFG в меню редактора для ввода нового
сегмента. Команды MEND, RET, RETI, LBL, SBR и INT требуют отдельных сегментов.

Общие указания

Примеры SIMATIC S7-200 предоставляются заказчику бесплатно. Данные примеры не привязаны к
конкретной задаче и являются общей информацией о возможностях применения S7-200. Решение
заказчика может отличаться от приведенного здесь.

За правильную работу системы заказчик несет ответственность сам. Мы обращаем Ваше внимание
на действующие нормы Вашей страны и предписания по установке соответствующей системы.
Ошибки и изменения возможны.




